
Leistungsfähigkeit war immer einer der wesentlichen Vorteile von FoxPro, insbesondere der Datenbankmaschine. Mit der Einführung von Visual FoxPro mit seiner Objektorientierung, seiner verbesserten Maschine und seinen Client/Server-Funktionen wird das Gesamtprodukt noch stärker. Die andere Seite der Medaille ist, daß das Produkt auch immer komplexer wird. Während also das Programmieren robuster, objektorientierter Anwendungen, die auch auf Daten aus entfernten Datenspeichern zugreifen können, immer einfacher wird, wird es leider auch immer einfacher, langsame Anwendungen zu erstellen. 


Die Intention dieses Textes ist es, praktische Tips und Methoden darzulegen, mit denen es Ihnen möglich ist, die Leistungsfähigkeit Ihrer Anwendungen zu verbessern. Durch die Verwendung dieser Techniken bei der Erstellung Ihrer Applikationen können Sie signifikante Verbesserungen bei der Benutzerschnittstelle und der Datenzugriffsgeschwindigkeit erzielen. 

Obwohl einige der Tips für manche von Ihnen offensichtlich sein werden, gehen wir davon aus, daß die gesamte Kollektion jedem etwas zu bieten hat, wenn es darum geht, die Gesamtleistung einer Anwendung zu steigern. Wir hoffen, daß diese Sammlung noch an Umfang gewinnen wird, und möchten alle auffordern, Ihren Beitrag dazu zu leisten und Ihre Ideen Ihren Kollegen der FoxPro-Gemeinde zugute kommen zu lassen. 

Hardwarekonfiguration 

In diesem Abschnitt werden Möglichkeiten vorgestellt, das Betriebssystem so anzupassen, daß es unter Visual FoxPro optimal läuft. In manchen Fällen sollte man ein wenig mit den Werten des Laufwerkscaches spielen, um das Optimum für die Anforderungen der eigenen Anwendung zu finden. 

Windows 3.x (VFP 3.0) 

· Falls Sie nur über 8 MB RAM verfügen, sollte der gesamte Laufwerkscache weniger als 1 MB betragen. 

· Falls SMARTDRV der einzige Laufwerkscache ist, den Sie verwenden, dann setzen Sie seine Größe auf einen Wert um die 512 kB. 

· Falls Sie Windows für Workgroups verwenden und den 32 bit Cache benutzen (im Abschnitt Virtueller Speicher der Systemsteuerung), sollten Sie diesen Cache auf weniger als 1 MB einstellen (512 kB genügen meist). Die Voreinstellung ist 4 MB. Setzen Sie, während Windows läuft, alle anderen Caches auf 0. 

Windows NT 
Im großen und ganzen arbeiten Windows NT und Visual FoxPro hervorragend zusammen. Das einzige Problem entsteht, wenn man VFP auf einem Windows NT Advanced Server betreibt. Der Advanced Server verwendet wesentlich mehr Speicher, als FoxPro erwartet, und VFP kann daher dazu tendieren, zuviel Speicher für den Eigengebrauch zu reservieren. 

Wenn Sie das verhindern wollen, sollten Sie die Systemfunktion SYS(3050) dazu verwenden, den Speicherbufferpool von VFP auf etwa ein Drittel des voreingestellten Wertes (siehe unten, Abschnitt Speichermanagement) einzurichten. Sie könnten beispielsweise den folgenden Code an Ihren Startupcode oder die Datei CONFIG. FPW anfügen: 

=SYS (3050, VAL (SYS (3050, 1, 0)) / 3) 

Lokale Daten

Rushmore und Indexe 
Im Kapitel 17 des Entwicklerhandbuchs finden Sie eine tiefergehende Abhandlung der Rushmoretechniken. Die folgenden Tips sind Ergänzungen. 

Indexe verwenden 

Es ist eine einfache Tatsache, daß die Verwendung von Indizes die Datenabfrage beschleunigt. Indizes in Visual FoxPro und den Vorgängerversionen sind Datenstrukturen, die auf einem Baummodell beruhen. Das Nachschlagen in der Indextabelle geschieht also selbst außerordentlich rasch. 

Einer der einzigartigen Aspekte von FoxPro ist die Unterstützung von auf Ausdrücken basierenden Indizes. Die meisten DBMS erlauben Indizes auf ein oder mehrere Felder; FoxPro gestattet, beliebige Ausdrücke, ja sogar benutzerdefinierte Funktionen, in Indizes zu verwenden. 

Sie sollten sich allerdings folgender Tatsache bewußt sein: Je mehr Indizes Sie einer Tabelle anfügen, desto langsamer wird der Vorgang des Erneuerns oder Einsetzens in die Tabelle vor sich gehen, da FoxPro jeden Index erneuern muß. Im allgemeinen sollten Sie einen Index nur auf Daten legen, die in Filtern oder Joins verwendet werden. 

Sie sollten es vermeiden, Indizes auf Felder zu legen, in denen lediglich einige wenige diskrete Werte vorkommen (wie z.B. in logischen Feldern). 

Wann werden Abfragen optimiert? 
Es ist wichtig, die Umstände zu kennen, unter denen Abfragen oder Filter auf Tabellen in VFP optimiert werden, und wann das nicht oder nur teilweise geschieht. 

Visual FoxPro optimiert Filterbedingungen, indem es nach Indexschlüsselbegriffen sucht, die der linken Seite der Filterbedingung entsprechen. Oft versuchen Entwickler irrtümlich, den Tagnamen eines Indexes einer Filterbedingung anzupassen, nur um dabei herauszufinden, daß VFP Abfragen nicht auf diese Weise optimieren kann. Es folgt nun ein Beispiel dafür, wie man Indizes nicht einsetzen sollte: 

USE CUSTOMERS
INDEX ON UPPER(cu_name) TAG name 

SELECT * FROM customers ; 
WHERE cu_name="BILL" && nicht optimiert. 
*Kein Index auf "cu_name" 

Die richtige Methode, die Vorteile eines Indexes in einer gefilterten Abfrage zu nutzen, ist: 

SELECT * FROM customers; 
WHERE UPPER(cu_name)="BILL" && Optimiert! 

Beschleunigen der Datenbankmaschine 
Primär- und Kandidatenschlüssel verwenden 

Die typische Vorgehensweise beim Anfügen von Datensätzen mit eindeutiger ID in FoxPro 2.x war, einen Index auf das Schlüsselfeld zu erstellen. Wenn dann ein Datensatz eingefügt werden sollte, startete man eine Prozedur, die SEEK gebrauchte, um festzustellen, ob die ID bereits in Verwendung war. Obwohl das schon recht schnell war, war es noch nicht optimal. 

Durch die Vergabe von Primär- und Kandidatenschlüsseln werden Eingaben oder Modifikationen in FoxPro-Tabellen von der Datenbankmaschine auf Eindeutigkeit untersucht. Dies geschieht auf einer tiefliegenden Ebene und ist extrem schnell. 

Internationales 

· Leistungssteigerung mit Unterscheidungszeichen: 

Wir haben in VFP die Art und Weise verbessert, in der das Nachschlagen von Daten, die internationale Zeichen (Zeichen mit Unterscheidungszeichen wie ä, é etc.) enthalten, optimiert wird. Falls das Ergebnis eines Indexausdrucks so ausfällt, daß es lediglich Zeichen enthält, die nie Unterscheidungszeichen verwenden (wie B oder M), wird das Nachschlagen dadurch beschleunigt. 

· Nichtmaschinelle Sortierfolgen: 

Es gibt verschiedene Gründe dafür, warum eine nichtmaschinelle Sortierfolge (wie General) langsamer ist als eine maschinelle Sortierfolge: 

Nichtmaschinelle Indexschlüssel sind doppelt so groß, um mögliche Unterscheidungszeichen aufnehmen zu können. 

Nichtmaschinelle Sortierungen verwenden viele spezielle Regeln für die Indizierung von Zeichen, um saubere Resultate liefern zu können. 

Da die maschinelle Sortierfolge schneller ist, zieht man sie im allgemeinen für Joins und Suchen vor, während andere Sortierfolgen für das Ordnen von Datensätzen perfekt geeignet sind. 

Beachten Sie bitte, daß VFP nur Indizes benutzt, die mit der aktuellen Einstellung von SET COLLATE erstellt wurden. Typischerweise umgeht man dies, indem man zwei Indizes auf die primären Such-/Sortierfelder legt. 

SET COLLATE TO "GENERAL" 
INDEX ON lastname TAG lastname && Sortierindex 
SET COLLATE TO "MACHINE" 
INDEX ON lastname TAG _lastname && Suchindex 

Wenn Sie nun also ein SEEK, SELECT oder JOIN auf das Feld lastname anwenden wollen und dabei den schnelleren maschinellen Index nutzen möchten, verwenden Sie vorher einfach SET COLLATE TO "MACHINE". Rushmore greift dann auf den in der Sortierfolge erstellten Index zu, und die Bearbeitung wird erheblich beschleunigt. 

· SQL SELECT: 
Wird ein SQL SELECT unter einer maschinellen Sortierfolge durchgeführt, so werden die Anweisungen ORDER BY und GROUP BY ebenfalls diese Sortierfolge verwenden. Falls Sie die resultierenden Datensätze aber in einer nichtmaschinellen Sortierfolge erhalten möchten, so können Sie entweder das SELECT unter der entsprechenden Sortierfolge ausführen - was langsamer sein könnte - oder aber das Problem in zwei Schritte zerlegen wie folgt: 

* Schritt 1: 
* Auswahl der Datensätze in der maschinellen Sortierfolge 

* Nur 'machine'-Indexe
SET COLLATE TO "MACHINE" 
SELECT * FROM table INTO TABLE temp1 ; 
WHERE lastname = "Muller" 

* Schritt 2: 
* Ordnen der Datensätze nach der Sortierfolge "general". 

* Nur 'general'-Indexe 
SET COLLATE TO "GENERAL" 
SELECT * FROM temp1 INTO TABLE output ; 
ORDER BY lastname 

Neue Datentypen verwenden 

Für Visual FoxPro wurden vier neue Datentypen eingeführt: DateTime, Integer, Double und Currency. Alle Daten, die diese Typen verwenden, werden als binäre Daten auf der Platte gespeichert. Integer ist hierbei ein 4 Byte Binärwert, die anderen sind 8 Byte Binärwerte. Das hat zwei Vorteile: Erstens ist der zum Speichern benötigte Raum auf der Platte geringer - eine achtstellige Zahl, die als Numeric gespeichert wird, benötigt 8 Bytes, während sie nur 4 Bytes braucht, wenn sie als Integer abgelegt wird. Dadurch beschleunigt sich auch der Zugriff auf Daten und Indizes, da mehr Daten in die Speicherseiten geladen werden können. Der zweite Vorteil liegt darin, daß keine Datenübersetzung mehr vonnöten ist. Intern repräsentiert FoxPro Integerzahlen als 4 Byte Binärwerte und muß Zahlen im Format Numeric erst übersetzen, da sie als ASCII auf der Platte gespeichert werden. Bei jedem Zugriff muß also erst von ASCII nach binär oder umgekehrt übersetzt werden. Für die neuen Datentypen ist eine solche Übersetzung nicht mehr notwendig, der Datenzugriff wird also beschleunigt. 

Von den vier neuen Datentypen ist der Typ Integer sicherlich der wichtigste, was die Geschwindigkeit anbetrifft. Wo auch immer es möglich ist, sollte man den Typ Integer für primäre und fremde Schlüsselwerte einsetzen. Die Vorteile sind kleinere DBFs, kleinere Indizes und (vielleicht am wichtigsten) wesentlich raschere Joins. 

Buffer und Transaktionen 

Transaktionen kurzhalten 
Transaktionen sollen kurz und bündig sein. Dazu das folgende Beispiel: 

BEGIN TRANSACTION
DO FORM BIGFORM TO nResult 
IF nResult = IDCOMMIT 
END TRANSACTION 
ELSE 
ROLLBACK 
ENDIF 

Was geschieht hier? Wenn Sie denken, daß dies eine brauchbare Vorgehensweise beim Umgeben einer Operation mit einer Transaktion sei, haben Sie recht. Allerdings sind Transaktionen so angelegt, daß sie Datenerneuerungen umgeben sollen und sonst nichts. Indem man sie mit mehr belastet - so wie oben gezeigt, mit Benutzerschnittstelle oder sogar mit Programmkonstruktionen wie CASE -, öffnet man der Ineffizienz beim eigentlichen Erneuern der Daten Tür und Tor. 

In Visual FoxPro ist dies signifikant, da der Gebrauch von Transaktionen dazu führt, daß Datensätze gesperrt werden. Während die Transaktion andauert und Daten erneuert werden, sind die entsprechenden Datensätze gesperrt. Diese werden erst freigegeben, wenn die Transaktion abgeschlossen oder zurückgenommen wird. Selbst wenn man RLOCK() oder FLOCK() gefolgt von UNLOCK einsetzt, bleiben die Datensätze gesperrt, bis entweder die Anweisung END TRANSACTION oder ROLLBACK erfolgt. 

Darüber hinaus muß VFP für das Anfügen von Datensätzen den Kopf der Tabelle sperren, und auch diese Sperre bleibt so lange bestehen, bis die Transaktion abgewickelt ist. In einem großen System mit vielen Anwendern wird es daher entscheidend, die Sperrfrist auf die Datensätze zu minimieren. Eine bessere Lösung für das oben angeführte Beispiel wäre etwa: 

* Sicherungsmethode aus dem Steuerknopf cmdSave 

BEGIN TRANSACTION 
UPDATE PRODUCTS SET reorder_amt = 0 ; 
WHERE discontinued = .T. 
END TRANSACTION 

SYS(3051) verwenden 

Diese SYS-Funktion steuert die Menge an Zeit in Millisekunden, die VFP wartet, bevor es nach einem Fehlversuch erneut versucht, einen Datensatz, eine Tabelle, einen Index oder ein Memofeld zu sperren. Die Voreinstellung beträgt 333 Millisekunden. 

Falls viele Sperren auf den Datenbankdateien zu erwarten sind, kann es lohnend sein, diesen Wert heraufzusetzen (das Maximum ist 1000), da die Anwendung keine Zeit damit vergeudet, Sperrverlangen über das Netzwerk zu schicken. 

Sind allerdings nur wenige Sperren zu erwarten, so sollten Sie diesen Wert niedriger ansetzen (das Minimum beträgt 100). So kann eine Sperroperation beschleunigt werden. 

Ferne Daten 

Das Ermitteln von Daten auf einem Back-End kostet. Um die Daten aus einer Datenbank auf dem Server zu holen, müssen die folgenden Schritte durchlaufen werden: 

· Der Client schickt die Abfrage an die Datenbank auf dem Back-End Server. 

· Der Server muß die Abfrage zerlegen und kompilieren. 

· Ein Resultatsatz wird erzeugt. 

· Der Server meldet dem Client, daß das Resultat fertiggestellt wurde. 

· Der Client empfängt die Daten über das Netzwerk vom Server. 

Es gibt viele verschiedene Vorgehensweisen, um die Ermittlung oder Erneuerung von Daten zu beschleunigen. 

Nur die benötigten Daten ermitteln 

Es ist wohl eher die Ausnahme, daß eine funktionale Einheit einer Anwendung (meistens eine Form oder ein Bericht) auf alle Daten einer Tabelle zugreifen muß. Erzeugt man also Fernabfragen, die lediglich die benötigten Felder und Reihen empfangen oder erneuern, so läßt sich dadurch der Aufwand an transferierten Daten optimieren. Hat man beispielsweise eine Fernabfrage zu einer Form mit vier Steuerelementen (Customer_ID, Company, Contact und Address aus der Tabelle Customers), dann könnte man die Abfrage mit dem folgenden SELECT-Statement ausführen: 

SELECT * FROM customers 

Das wäre leider ziemlich ineffizient, da man so weitaus mehr Felder erhält, als man eigentlich braucht. Besser wäre es da, wenn das SELECT-Statement folgendermaßen aussähe: 

SELECT customer_id, company, contact, address ; 
FROM customers 

Verwenden einer WHERE Klausel 

Der richtige Gebrauch einer WHERE Klausel kann den Datenzugriff weiter beschleunigen, indem die ermittelte oder erneuerte Datenmenge weiter eingeschränkt wird. Bleiben wir bei dem obigen Beispiel. Jetzt wollen wir nur die Kunden aus der Nordwestregion (der USA). Das SELECT könnte nun aussehen wie folgt: 

SELECT customer_id, company, contact, address ; 
FROM customers ; 
WHERE region = 'NORTHWEST' 
WHERE Klauseln parametrisieren 

Die Flexibilität der Views unter Visual FoxPro und die Pass Through Technologie von SQL ermöglichen es Ihnen, in der WHERE Klausel für SELECT, UPDATE und DELETE Laufzeitparameter zu verwenden. Um im bereits bewährten Beispiel zu bleiben: Man kann Informationen zu jeder Region ermitteln, wobei die Region jeweils zur Laufzeit festgelegt wird: 

SELECT customer_id, company, contact, address ; 
FROM customers ; 
WHERE region = ?pcRegion 

Hierbei ist pcRegion der Name eines Parameters. Dieser sollte bereits existieren (er muß es nicht), wenn die Abfrage gestartet wird, damit nur die benötigte Information geliefert wird. 

Korrekte Update-Kriterien verwenden 

Die Update-Kriterien werden in einer Tabelle im View Designer abgelegt. Dort können Sie angeben, wie die Daten erneuert, eingesetzt oder gelöscht werden sollen. Der Abschnitt der Tabelle, der mit "SQL WHERE clause includes" bezeichnet ist, dient dazu, den Inhalt der WHERE Klauseln der Operationen UPDATE sowie DELETE festzulegen. Dies ist wichtig, um Datenkonflikte auf dem Back-End zu vermeiden, und dient auch der Beschleunigung der Bearbeitung. Nehmen wir wieder das View von oben und gehen davon aus, daß Customer_ID das Schlüsselfeld sei. Weiterhin wollen wir annehmen: Alle Felder mit Ausnahme des Schlüsselfeldes sollen erneuert werden können. Anderes wäre auch selten, da eine Änderung an Schlüsselfeldern ein DELETE gefolgt von einem INSERT impliziert. Die Felder, um die es geht, sind: Company, Contact und Address. Ihre WHERE Klausel hinge dann von den in der Optionsgruppe SQL WHERE Clause Includes gewählten Optionen ab: 

Aus der nachfolgenden Tabelle sollte ersichtlich werden, daß die verwendete WHERE Klausel einen grossen Einfluß auf die Leistung haben kann, insbesondere, wenn viele Felder geändert wurden (im Fall 2: Schlüsselfeld und Modifizierte Felder) oder viele Felder zu erneuern sind (im Fall 3: Schlüsselfeld und Erneuerbare Felder). Dies liegt daran, daß VFP die Werte im Datenbuffer Oldval (der nur ein Systemcursor ist) mit den Werten auf der Platte (der Serverdatenbank) vergleichen muß, damit es feststellen kann, ob sich ein Update-Konflikt zugetragen hat, d.h., ob ein anderer Benutzer oder Prozeß die Daten verändert hat, seit diese an Sie überwiesen wurden. Dieser Vergleichsvorgang kann recht zeitaufwendig sein, wenn eine grössere Update-Operation durchgeführt wird. 

	Einstellung
	Resultierende WHERE Klausel

	Key Fields Only
	WHERE OLDVAL (customer_id) = CURVAL (custmer_id)

	Key and Modifield
Fields (Voreinstellung)
	WHERE OLDVAL (customer_id) = CURVAL (custmer_id) AND 
OLDVAL (<first_modifield_field>) = CURVAL (<first_modifield_field>) AND OLDVAL (<next_modifield_field>) AND etc.

	Key and Updatable
Fields
	WHERE OLDVAL (customer_id) = CURVAL (custmer_id) AND 
OLDVAL (company) = CURVAL (company) AND 
OLDVAL (contact) = CURVAL (contact) AND
OLDVAL (address) = CURVAL (address)

	Key and Timestamp
	WHERE OLDVAL (customer_id) = CURVAL (custmer_id) AND
OLDVAL (timerstamp) = CURVAL (timerstamp)


Anmerkung: Diese Einstellungen können auch durch die Eigenschaft WhereType für Views festgelegt werden. Dies geschieht über die Funktionen DBSETPROP() oder CURSORSETPROP(). 

Wir empfehlen daher, daß Sie die Einstellung "Schlüsselfelder und Modifizierte Felder" in den meisten Fällen einsetzen, da die Felder, die Sie auf dem Back-End erneuern immer eine Untermenge der überhaupt erneuerbaren und erst recht eine Untermenge aller Felder des View sind. 

Bei Serverdatenbanken, die Timestamps unterstützen, empfehlen wir Ihnen, die Einstellung 4 - "Schlüsselfelder und Timestamps" - zu verwenden. Dies geht sogar noch vor "Schlüsselfelder und Modifizierte Felder". Ein Timestamp ist eine Version einer Zeile und wird jedesmal erneuert, wenn eine Änderung in der Zeile vorgenommen wird (Nebenbei bemerkt impliziert der Datentyp Timestamp nicht unbedingt Datum oder Zeit, sondern nur einen numerischen Wert. Sie müssen sich daher keine Gedanken darüber machen, die Systemuhren der verschiedenen Maschinen könnten zu Problemen beim Updaten führen). Beim Gebrauch dieser Einstellung wird lediglich eine Spalte in der WHERE Klausel verglichen, daher ist dieser Vorgang schneller bearbeitet. 

Verwenden der Eigenschaft BatchUpdateCount 

Manche Serverdatenbanken (wie der SQL Server von Microsoft) gestatten den Einsatz von Statementstapeln. Diese können als ein Paket an den Server geschickt werden. Setzt man beispielsweise den Wert dieser Eigenschaft auf 4 und hat an den ersten 10 Datensätzen Veränderungen vorgenommen, so lauten die von VFP an die Serverdatenbank gesandten SQL Statements wie folgt: 

UPDATE customer SET contact = 'John Jones' ; 
WHERE customer_id = 1;
UPDATE customer SET contact = 'Sally Park' ; 
WHERE customer_id = 2; 
UPDATE customer SET company = 'John Jones' 
WHERE customer_id = 3; 
UPDATE customer SET contact = 'John Jones' 
WHERE customer_id = 4 

Diese Vorgehensweise beschleunigt Datenzugriffe schon dadurch, daß die reine Anzahl an Datenpaketen durch das Netzwerk drastisch reduziert wird. Außerdem können die Statements von der Serverdatenbank als Stapel kompiliert werden anstatt stückweise. 

Sie sollten mit verschiedenen Werten für diese Eigenschaft und die Eigenschaft PacketSize herumexperimentieren, um die optimalen Werte für Ihre Updates zu finden. 

Diese Eigenschaft kann im Dialog Advanced Options des Abfragemenüs eingestellt werden (oder im Code durch die Funktionen DBSETPROP() or CURSORSETPROP() ). 

Eigenschaft PacketSize verwenden 

Diese Eigenschaft steuert die Größe eines Netzwerkpakets in Bytes, wie es im Verkehr mit dem Server Verwendung findet. Sie wird durch ODBC gesetzt und kann jeden nichtnegativen Wert annehmen. Verschiedene Netzwerkbetreiber gehen unterschiedlich auf diese Eigenschaft ein. Sie sollten daher die Dokumentation des Netzwerks zu Rate ziehen. Novell Netware beispielsweise benutzt eine maximale Paketgröße von 512 Bytes. Setzt man die Eigenschaft PacketSize also auf einen größeren Wert, wird dies keinen weiteren Vorteil bringen. 

Die Voreinstellung dieser Eigenschaft ist 4096 Bytes. Falls Ihr Netzwerk größere Pakete zuläßt, können Sie diesen Wert noch erhöhen, um einen maximalen Durchsatz pro Abfrage (SELECT, INSERT, UPDATE, DELETE) zu erzielen. 

Nachschlagetabellen lokal halten 

Bei vielen Gelegenheiten kommt es vor, daß die Anwendung sehr häufig auf Daten, die nur gelesen werden können, zugreift. Beispiele hierfür wären eine Tabelle der Bundesstaaten oder eine Tabelle der Titel der Angestellten, welche man in bestimmten Anwendungen (besonders in den USA) oft einsetzt. Indem man diese Daten lokal hält, d.h., nicht immer vom Server laden muß, wird das Nachschlagen dieser Informationen sehr beschleunigt. Diese Vorgehensweise ist für alle Arten von Daten sinnvoll, die nicht oder nur sehr selten geändert werden. 

Lokale Regeln verwenden 

Daß VFP Regeln auf Feld- und Datensatzebene für lokale und ferne Views unterstützt, ist eine wenig beachtete Tatsache. Diese Regeln lassen sich dazu verwenden, zu verhindern, daß Daten, die nicht zu anderen Daten oder zu Geschäftsregeln passen, in die Datenbank gelangen. Der Vorteil, der darin liegt, diese Regeln in das View anstatt in die Back-End-Tabelle einzubauen, liegt darin, daß man die ungültigen Daten aufspüren kann, bevor sie über die Leitung gehen und dort den Verkehr aufhalten. Der Nachteil ist allerdings, daß diese Regeln nicht automatisch zu den Regeln der Tabellen auf dem Back-End Server passen. Verändert man also die Definition der Tabellen, so muß man die Regeln in den lokalen Viewdefinitionen in den meisten Fällen ändern. Sind diese Regeln einfach, ist das zum Glück keine allzu große Last. Außerdem ändern sich die Regeldefinitionen nicht allzu häufig, daher muß man auch nicht damit rechnen, die lokalen Regeln oft ändern zu müssen. 

Form/Objekt-Performance 

In diesem Abschnitt werden Sie Tips finden, wie man die Performance von Objekten - mit besonderem Augenmerk auf Formen - innerhalb einer Anwendung erhöhen kann. 

Datenumgebung verwenden 

Die benötigte Zeit für das Öffnen einer Tabelle läßt sich durch die Verwendung der Datenumgebung des Form bzw. Berichtsdesigners gegenüber der Zeit erheblich reduzieren, die zum Ausführen der Anweisungen USE, SET ORDER und SET RELATION im Ereignis Laden der Form benötigt wird. Dies liegt daran, daß VFP Maschinenaufrufe auf unterer Ebene einsetzt, um die Tabellen zu öffnen und Indizes und Relationen einzurichten. 

Anzahl der Formen in einem Formset begrenzen 

Man sollte Formsets nur dann einsetzen, wenn man wirklich will, daß sich eine Gruppe von Formen private Daten einer Session teilt. Beim Betreiben eines Formsets wird VFP im wesentlichen aufgetragen, alle Forminstanzen im Formset und alle Steuerelemente aller Formen zu erzeugen, auch wenn nur die erste Form des Formsets überhaupt angezeigt werden soll. Das kann natürlich sehr zeitaufwendig werden und ist eigentlich immer unnötig, wenn nicht private Daten einer Session geteilt werden müssen. Die bessere Lösung liegt darin, DO FORM einzusetzen, wenn man andere Formen benötigt. 

Andererseits erhöht sich die Leistung beim Zugriff auf die anderen Formen eines Formsets - falls man einen solchen verwendet -, da diese ja schon geladen sind, auch wenn sie nicht angezeigt werden. 

Pageframe Steuerelemente dynamisch laden 

Genau wie Formsets laden Pageframes die Steuerelemente für jede Seite, wenn das Pageframe geladen wird. Ein probates Mittel ist hier, eigene Klassen aus den Steuerelementen der einzelnen Seiten des Pageframes zu erstellen. 

Das Ziel dabei soll es sein, die Instanzierung von Steuerelementen nicht dargestellter Seiten so lange zu verschieben, bis ein Anwender auf diese Seiten zugreift. Folgende Vorgehensweise ist dabei einzuhalten: 

1. Erstellen Sie Ihre Form genauso wie sonst auch einschließlich aller Steuerelemente auf allen Seiten. 

2. Wenn Sie mit der Erstellung fertig sind, gehen Sie zur zweiten Seite und speichern die dortigen Steuerelemente als Klasse. 

3. Öffnen Sie die von Ihnen erzeugte Klasse, und kontrollieren Sie, ob die Steuerelemente immer noch das richtige Layout haben. 

4. Wiederholen Sie die Schritte 2. und 3. für die dritte und alle weiteren Seiten des Pageframes. 

5. Plazieren Sie den folgenden oder einen analogen Code in der Activate-Methode der zweiten und aller weiteren Seiten des Pageframes: 

IF This.ControlCount=0 
This.Addobject("cnrpage1","cnrpage1") 
This.cnrpage1.visible=.T. 
ENDIF 

Mit Hilfe dieser Vorgehensweise können Sie das Laden der Formen erheblich beschleunigen, da die Steuerelemente der zweiten und aller weiteren Seiten des Pageframes erst geladen werden, wenn sie wirklich benötigt werden. 

Steuerelemente dynamisch an Daten anbinden 

Eine weitere Leistungssteigerung läßt sich bei Formen erzielen, die viele Steuerelemente verwenden, die an Daten angebunden sind. Diese Anbindung zögert man einfach so lange hinaus, bis sie wirklich gebraucht wird. 

Sie können die Tabellen und Views Ihrer Form in die Datenumgebung einstellen, damit diese bereits geöffnet werden, wenn die Form geladen wird. Wird dann ein Steuerelement, sagen wir eine Combobox, aufgerufen, bindet man das Steuerelement an den Datenwert an. Im untenstehenden Beispiel habe ich eine Combobox mit dem Feld Customer.Company im Ereignis GotFocus verknüpft: 

* Überprüft, ob das Steuerelement bereits angebunden. 
IF this.RecordSource = "" 

* Setzt RecordSource auf richtigen Wert 
* und setzt RecordSourceType auf "Felder" 
this.RecordSource = "customer.company" 
this.RecordSourceType = 6 

* Steuerelement Erneuern 
this.Refresh 
ENDIF 

Mittels dieser Vorgehensweise für datenverbundene Steuerelemente läßt sich die Ladezeit für Formen dramatisch verringern. 

LockScreen verwenden 

Diese Eigenschaft ermöglicht es Ihnen, Änderungen an der Darstellung auf dem Bildschirm nach der Modifikation von Steuerelementen einer Form zu verzögern. Macht man beispielsweise Steuerelemente sichtbar oder unsichtbar, ändert die Farben oder bewegt Datensätze in angebundenen Steuerelementen ist es wesentlich effizienter, die Übertragung der Änderungen auf den Bildschirm so lange hinauszuzögern, bis sie alle abgeschlossen sind. 

thisform.LockScreen = .T. 
thisform.MyButton.Caption = "Save" 
thisform.MyGrid.BackColor = RGB (255, 0, 0) && Rot 
SKIP IN customers 
SKIP IN orders 
thisform.Refresh 
thisform.LockScreen = .F. 

In diesem Beispiel habe ich eine Bezeichnung eingesetzt, eine Hintergrundfarbe geändert und Datensätze in allen Steuerelementen bewegt, die an Kunden und Aufträge angebunden sind. Ohne die Eigenschaft LockScreen würde man sehen können, wie jeder dieser Vorgänge eine Zeichenoperation auf die betroffenen Steuerelemente anwendet, und die Gesamtperformance wäre alles andere als großartig. 

Die hier vorgestellte Methode bringt man typischerweise dann zur Anwendung, wenn mehrere Änderungen an der Darstellung einer Form gleichzeitig anstehen. Es ist sicher nicht ratsam, LockScreen bei jeder Änderung der Bildschirmdarstellung zu verwenden, sondern wirklich nur bei einer Gruppe von Änderungen wie oben gezeigt. 

WITH … ENDWITH 
oder Objektreferenzen verwenden 

Wenn Sie auf eine Eigenschaft eines Objekts mittels der Syntax "Objekt.Eigenschaft" verweisen, muß Visual FoxPro zunächst nach dem Objekt suchen, bevor es auf die Eigenschaft zugreifen kann. Der folgende Code soll dies illustrieren. Hier wird VFP dazu gebracht, sich durch vier Objekte hindurchzuhangeln (thisform, pgfCustInfo, pagCustName und cboName), um die Eigenschaft zu finden, die gesetzt werden soll. Dies geschieht dreimal: 

thisform.pgfCustInfo.pagCustName.txtName.Value = ; 
"Fred Smith" 
thisform.pgfCustInfo.pagCustName.lblName.Caption = ; 
"Name" 
thisform.pgfCustInfo.pagCustName.grdOrders.BackColor = ; 
RGB (0,0,0) 

Wenn Sie mehr als eine Eigenschaft eines Objekts aufeinanderfolgend ändern wollen, sollten Sie WITH ... ENDWITH verwenden. Dies führt dazu, daß VFP das entsprechende Objekt nur einmal suchen und finden muß. Die Einstellungen aus obigem Beispiel lassen sich so wesentlich rascher umsetzen: 

WITH thisform.pgfCustInfo.pagCustName 
.txtName.Value = "Fred Smith" 
.lblName.Caption = "Name" 
.grdOrders.BackColor = RGB (0,0,0) 
ENDWITH 

Eine andere Möglichkeit, dieses Problem anzugehen, ist der Gebrauch von Objektreferenzen. Objektreferenzen sind nichts weiter als Speichervariablen, die einen Verweis auf ein Objekt enthalten. Ausgehend von obigem Beispiel, könnte man also auch folgendermassen vorgehen: 

oPageFrame = thisform.pgfCustInfo.pagCustName 
oPageFrame.txtName.Value = "Fred Smith" 
oPageFrame.lblName.Caption = "Name" 
oPageFrame.grdOrders.BackColor = RGB (0,0,0)[GFGI1] 

Das oben eingeführte Beispiel würde sich auf diese Art vielleicht nicht allzusehr beschleunigen lassen, aber wenn das Objekt (in diesem Fall das Seitenrahmenobjekt pagCustName) in Ihrer Anwendung häufiger aufgerufen wird oder gar in einer Schlaufe steht, so wird die Geschwindigkeit sich erhöhen. 

Speichervariablen für mehrfache Aufrufe verwenden 

Auf die komplette <Container.Objekt.Eigenschaft> zuzugreifen ist langsamer, als auf eine Speichervariable (Memvar) zuzugreifen. Dies kann sich innerhalb von Programmschlaufen besonders unangenehm bemerkbar machen. Die übliche objektorientierte Praxis ist daher, den Inhalt der Eigenschaft eines Objekts in eine Speichervariable abzulegen, wenn die Eigenschaft häufig benötigt wird. Der folgende Code füllt beispielsweise ein Eigenschaftsarray: 

FOR nCounter = 1 to 256 
thisform.aCharArray[x] = ; 
SUBSTR (thisform.cCharString, x, 1) 
ENDFOR 

Der folgende Code ist, obwohl umfangreicher, schneller, da der Teil mit der <Objekt.Eigenschaft>-Syntax aus der Schlaufe genommen wurde. 

* Kopiert String auf lokale Variable 
lcChar = thisform.cCharString 
* Erzeugt lokales Array 
LOCAL laCharArray[256 
FOR nCounter = 1 to 256 
laCharArray[x] = SUBSTR (laChar, x, 1) 
ENDFOR ^ 

* Kopiert lokales Array auf Eigenschaftsarray 
=ACOPY (laCharArray, thisform.aCharArray) 

Große Mengen an Xbase-Code in Init-/Refresh-/Paint-Methoden vermeiden 

Diese Ereignisse kommen sehr oft vor. Sie sollten die Menge an Code, die Sie hier einsetzen, gering halten. Dies gilt besonders für Refresh- und Paint-Methoden, da diese besonders häufig sind. 

Verlagern Sie Initcode auf weniger häufig benutzte Methoden wie Activate, Click und GotFocus. Verwenden Sie eine Eigenschaft des Steuerelements, um festzustellen, ob das Steuerelement bereits Code gestartet hat, der nur einmal aktiviert werden muß. 

Eigenschaft NoDataOnLoad bei Views verwenden 

Die Eigenschaft NoDataOnLoad in einem Datenumgebungscursor für ein View ist identisch mit der NODATA-Klausel in der USE-Anweisung für Views. Sie bewirkt, daß das View geöffnet wird, ohne wirklich Daten zu sammeln. Dies gilt sowohl für lokale wie Fernabfragen. 

Diese Vorgehensweise wird typischerweise in Verbindung mit parametrisierten Abfragen in der Datenumgebung angewandt (siehe Abschnitt über Parametrisierung von WHEN-Klauseln). Hätte man beispielsweise eine Form mit Kundeninformationen, so könnte man eine Abfrage verwenden, die auf Customer_ID parametrisiert ist. Sie geben dann eine gültige Kunden-ID ein und drücken einen Suchknopf. Der Code des Suchknopfs würde dann aussehen wie folgt (ich gehe davon aus, daß der Parameter für die Abfrage die Bezeichnung ThisForm.pnValue trägt, wobei pn eine ungarische Notation ist, die ich verwende, um eine Variable als numerischen Parameter zu bezeichnen): 

* Setzt Abfrageparameter (eine benutzerdefinierte 
* Eigenschaft die ich erzeugt habe) auf den Wert in der 
* Textbox CustomerID 
thisform.pnValue = thisform.txtCustomerID 

* Sperrt Bildschirm, um Darstellung zu verzögern
thisform.LockScreen = .T. 

* Requery auf das View freigeben 
=REQUERY ('customerview') 

* Refresh der datengebundenen Steuerelemente 
=thisform.Refresh() 

* Entfernt Bildschirmsperre 
thisform.LockScreen = .F. 

Der Code ist natürlich nur ein einfaches Beispiel, aber er ist typisch für die Arbeit mit parametrisierten lokalen oder Fernabfragen. 

Der Vorteil liegt hier darin, daß die Ladezeit der Form erheblich reduziert werden kann, da das View nicht dazu führt, daß irgendwelche Daten zu Ihnen, dem Client, zurückgebracht werden. Und trotzdem werden die Steuerelemente, die an Daten angebunden sein sollen, noch mit diesen Daten verbunden, da es ja einen offenen Arbeitsbereich gibt (nur ohne Daten darin). 

OLE-Performance 

OLE-Server vor Zugriff auf OLE Daten starten 

Steuerelemente, die an generische Felder angebunden sind, werden im allgemeinen schneller arbeiten, wenn die Server für diese Datentypen (z.B. Word oder Excel) bereits auf dem Clientrechner gestartet wurden. 

OLE-Automatisierung 

GetObject anstelle von CreateObject verwenden 

In manchen Fällen starten Automatisierungsserver (wie Excel) eine neue Instanz ihrer selbst, sogar wenn bereits eine läuft. Um dies zu verhindern und somit die Leistung zu erhöhen, sollte man die Funktion GetObject anstelle von CreateObject verwenden. Nehmen wir zum Beispiel den folgenden Aufruf: 

x = GetObject (, "excel.Application") 

So wird immer nur eine bereits vorhandene Instanz verwendet, wenn eine solche existiert. Hingegen wird 

x = CreateObject ("excel.Application") 

eine neue Instanz erzeugen. 

Lange Ausdrücke nicht wiederholt berechnen 

Das wiederholte Ausführenlassen von Ausdrücken, die Objekte innerhalb des OLE-Servers verwenden, kann zeitaufwendig werden. Es ist wesentlich effizienter, Unterobjekte in Speichervariablen zwischenzuspeichern (siehe Abschnitt über Speichervariablen bei mehrfachem Zugriff). 

Objekte "Als Icon" einsetzen 

Setzt man ein OLE-Objekt in ein Feld ein, kann man es als Icon - quasi ein Platzhalter - statt als ganzes Objekt einstellen. So läßt sich der Speicherverbrauch reduzieren, da VFP ein Präsentationsbild zusammen mit dem Objekt speichert, das viel Speicher belegen kann. Außerdem verringert sich die Zeit, die benötigt wird, um die Bildschirmdarstellung des Objekts zu erstellen, da ja nur ein Icon gerendert werden muß. 

Image-Steuerelemente verwenden 

Wenn einzelne Bitmaps wie etwa ein Firmenlogo in einer Anwendung verwendet werden sollen, sind Image-Steuerelemente bedeutend effektiver als OLEBoundControls. 

Möglichst manuelle Links verwenden 

Manuelle Links zu Objekten sind schneller, da automatische Links Zeit für die Meldung benötigen und da der OLE-Server nicht gestartet werden muß, um das Objekt zu zeichnen. Wenn Sie also ein Objekt nicht häufig verändern müssen, sollten Sie manuelle Links einsetzen. 

Speichermanagement 

SYS(3050) verwenden 

Mittels dieser SYS-Funktion ist es möglich, festzulegen, wieviel Vorder- und Hintergrundspeicher für Datenbuffer verwendet werden sollen. Vordergrundspeicher ist der Speicher, der für VFP verfügbar ist, wenn es im Vordergrund läuft. Hintergrundspeicher ist folgerichtig der Speicher, der VFP zur Verfügung steht, wenn es eine der im Hintergrund laufenden Anwendungen ist. 

Sie sollten mit den Werten ein wenig herumexperimentieren, um herauszufinden, welche Werte Ihre Anwendung am ehesten optimieren. 

MEMLIMIT nicht mehr verwenden 

MEMLIMIT wird von VFP nicht mehr erkannt. In FoxPro 2.6 diente MEMLIMIT als Konfigurationseinstellung dazu, den maximalen Speicher festzulegen, den FoxPro für seinen Gebrauch in Anspruch nahm. Dies können Sie so nun nicht mehr tun. Verwenden Sie stattdessen die Funktion SYS(3050), falls nötig. 

